Problem:
Range Minimum Query (RMQ)

Write a program which manipulates a sequence A = {a0, a1, . . . , an-1} with the following operations:

find(s, t): report the minimum element in as, as+1, . . . ,at.

update(i, x): change ai to x.

Note that the initial values of ai (i = 0, 1, . . . , n−1) are 231-1.

Input

n q
com0 x0 y0
com1 x1 y1
...
comq−1 xq−1 yq−1

In the first line, n (the number of elements in A) and q (the number of queries) are given. Then, q queries are given where com represents the type of queries. '0' denotes update(xi, yi) and '1' denotes find(xi, yi).

Output

For each find operation, print the minimum element.

Constraints

1 ≤ n ≤ 100000

1 ≤ q ≤ 100000

If comi is 0, then 0 ≤ xi < n, 0 ≤ yi < 231-1.

If comi is 1, then 0 ≤ xi < n, 0 ≤ yi < n.

Sample Input 1

3 5
0 0 1
0 1 2
0 2 3
1 0 2
1 1 2

Sample Output 1

1
2

Sample Input 2

1 3
1 0 0
0 0 5
1 0 0

Sample Output 2

2147483647
5

Solution in Haskell (judge verdict: Accepted)

import Data.Monoid
import Control.Monad
import Data.List
import Control.Applicative

instance Monoid Int where
    mempty = 2 ^ 31 - 1
    mappend = min

data Segtree m = Leaf {-# UNPACK #-} !m
               | Node
                    {-# UNPACK #-} !(Segtree m)
                    {-# UNPACK #-} !(Segtree m)
                    {-# UNPACK #-} !m
                    {-# UNPACK #-} !Int
                deriving (Show, Eq)

{-# INLINE size #-}
size :: Monoid m => Segtree m -> Int
size (Leaf _) = 1
size (Node _ _ _ n) = n

{-# INLINE value #-}
value :: Monoid m => Segtree m -> m
value (Leaf x) = x
value (Node _ _ x _) = x

combine :: Monoid m => Segtree m -> Segtree m -> Segtree m
combine lch rch = Node lch rch (value lch <> value rch) (size lch + size rch)

rangeQuery :: Monoid m => Int -> Int -> Segtree m -> m
rangeQuery a b (Leaf x)
    | b <= 0 || a >= 1 = mempty
    | otherwise = x
rangeQuery a b (Node lch rch x n)
    | b <= 0 || a >= n = mempty
    | a <= 0 && n <= b = x
    | otherwise = rangeQuery a b lch <> rangeQuery (a - lsize) (b - lsize) rch
        where lsize = size lch

update :: Monoid m => Int -> m -> Segtree m -> Segtree m
update k x (Leaf _) = Leaf x
update k x (Node lch rch _ _)
    | k < size lch = combine (update k x lch) rch
    | otherwise = combine lch $ update (k - size lch) x rch

build :: Monoid m => Int -> Segtree m
build n
    | n <= 1 = Leaf mempty
    | otherwise = combine (build (n `div` 2)) (build (n - n `div` 2))

processQuery :: Int -> [[Int]] -> [Int]
processQuery n = reverse . snd . foldl' step (build n, [])
    where step (tree, res) [0, x, y] = (update x y tree, res)
          step (tree, res) [_, x, y] = (tree, rangeQuery x (y + 1) tree : res)

splitRead = map read . words <$> getLine

main = do
    [n, q] <- splitRead
    replicateM q splitRead >>= mapM_ print . processQuery n